スイープ加振とは振動実験で良く行われている方法で、加振力の変動周波数を時間とともに変化させながら対象とする振動系の動特性を把握（特に、共振周波数）するものである。これを Modelica を用いて数値実験行うことを考える。基本となる振動系は図 1、Modelica スクリプトは図 2 であるが入力が異なってくる。図 1 の場合は一定加振周波数なので加振力は $d\cos\omega t$ で表現できる。ここに、ωが加振周波数である。スイープ加振の場合は、ωt の部分を何らかの形で置き換える必要がある。ここに、ωt が位相で、ωが周波数で両者の間には、位相を時間微分したものが周波数、もしくは周波数を時間積分したものが位相である点に注目していただきたい。スイープ加振の場合には、周波数を時間とともに変化させたい、例えば、時間で線形に周波数を上げて行きたい場合には、周波数は at となる。ここに、a は周波数の上昇速度である。従って、スイープ加振の際の位相はこれを時間積分したものであるから $at^2/2$ となる。



図 1　強制振動を受ける 1 自由度系ばね・ダンパ・マスモデル

$$m\frac{d^2x}{dt^2}+c\frac{dx}{dt}+kx=d\cos(\omega t)$$

```
model forcedVib
 Modelica.SIunits.Position x(start=0.0, fixed=true);
 Modelica.SIunits.Velocity v(start=0.0);
 Modelica.SIunits.Acceleration a;

 Modelica.SIunits.Frequency force_freq;
 Modelica.SIunits.Frequency resonance_freq;
 Modelica.SIunits.Force force;

 parameter Modelica.SIunits.Mass m=1.0;
 parameter Modelica.SIunits.TranslationalDampingConstant c=1.0;
 parameter Modelica.SIunits.TranslationalSpringConstant k=10000.0;
 parameter Real d=100000.0;
 parameter Modelica.SIunits.AngularFrequency omg=300.0;

equation
 v=der(x);
 a=der(v);
 m*a + c*v + k*x = force;
 force = d*cos(omg*time);
 force_freq = omg/2/Modelica.Constants.pi;
 resonance_freq = 1/2/Modelica.Constants.pi*sqrt(k/m);
end forcedVib;
```

図 2　図 1 の強制振動系を表現した Modelica スクリプト

以上の関係を図 3 に示す。周波数の変化形式がどのようなものであれ、この関係式を用いてスイープ加振の位相を定義できる。



図 3　変動加振の位相と周波数の関係

以上の知見を元に、線形で周波数ゼロから周波数を上昇させる場合のスイープ加振を表現する運動方程式は次式となる。

$$m\,\frac{d^2x}{dt^2} + c\frac{dx}{dt} + kx = d\cos\left(2\pi\frac{1}{2}at^2\right) \qquad (1)$$

図 3 では周波数を角周波数（rad）で考えたが、振動実験では一般にいわゆる周波数（ 1 秒間当たりのサイクル数）で表現する場合が多いので、これを考慮して()内は 2 π を乗じている。

式(1)を Modelica で直接表現すると図 4 となる。ここでは、a=1 とし、計算時間は 30 秒までとしている。なお、時間の設定は図 4 のシミュレーションのセットアップ画面の開始時刻を 0(s)、終了時刻を 30(s)とすればよい。すなわち、図 4 は加振周波数を 0Hz から 30Hz まで一秒間に 1Hz の上昇速度で 30 秒かけて、30Hz まで昇速させることを表現している。図 5 に計算例を示す。上図に、加振周波数、系の固有振動数および加振力の時間波形を、下図にその時のマストの応答振幅を示す。このように、加振周波数が系の固有振動数に近くなった時に、応答振幅が大きくなることが分かる。これがいわゆる共振現象である。スイープ加振の場合には、周波数が時々刻々変化しているため、理論的な共振周波数を時間的に若干過ぎてから応答振幅が最大となる。図 5 の結果もこの現象を捉えている。スイープ加振の周波数の変化速度を小さくすればするほど、応答振幅が最大となる加振周波数は理論上の共振周波数に近づいていく。

$$m\frac{d^2x}{dt^2}+c\frac{dx}{dt}+\ kx=\ d\cos(2\pi\frac{1}{2}at^2)$$




```
model forcedVibSweep
 Real x(start=0.0, fixed=true);
 Real v(start=0.0);
 Real a;
 Modelica.SIunits.Frequency force_freq;
 Modelica.SIunits.Frequency resonance_freq;
 Modelica.SIunits.Force force;
 parameter Real m=1.0;
 parameter Real c=1.0;
 parameter Real k=10000.0;
 parameter Real d=1.0;
 parameter Real CF=1.0;
equation
 v=der(x);
 a=der(v);
 m*a + c*v + k*x = force;
 force = d*cos(2*(1/2)*CF*time*Modelica.Constants.pi*time);
 force_freq = CF*time;
 resonance_freq = 1/2/Modelica.Constants.pi*sqrt(k/m);
end forcedVibSweep;
```

図 4　スイープ加振を表現した Modelica スクリプト



図 5　図 4 の計算例